

شرح الدالة SumProduct

الدالة SumProduct

تعتبر هذه الدالة من أقوى دوال الإكسيل و هي الدالة الوحيدة التي يمكن استخدامها بدل معادلات الصنف في معظم الحالات و السبب في ذلك قدرتها على التعامل مع الصفوفة Array.

و هذه القدرة غير متوفرة في الدوال الأخرى بشكل منفرد.

و الشكل العام لهذه الدالة هو

(...,SUMPRODUCT(array1,array2,array3=

فكما نلاحظ تأخذ المتغيرات كمصفوفات مرقمة من الرقم 1 و الحد الأعلى لها 255 مصفوفة .

و بشكل عام تقوم هذه الدالة بضرب الصفوف في المصفوفة كل حسب ترتيبه و من ثم تقوم بجمع حاصل الضرب و بالمثل يتضح الصورة إن شاء الله

لو افترضنا وجود القيم التالية في المدى A1:A4

العدد

3

4

2

و في المدى B1:B4 توجد البيانات التالية

السعر

2

1

5

العدد	السعر
3	2
4	1
2	5

و نرغب في جمع حاصل ضرب كل عدد في المدى الاول بالسعر الذي يقابله

بمعنى أن تكون العملية بشكلها النهائية هكذا :

2*3

1*4

5*2

و الحل بالطرق العادية سيكون بوضع معادلة في العمود C و من ثم سحبها للأسفل و من ثم جمع حاصل عملية الضرب و هذا الإجراء في حالات الجداول الكبيرة غير منطقي, من هنا يمكننا أن نستكشف الخطوات الأولى في التعامل مع الدالة

SUMPRODUCT

و تكون المعادلة بالشكل التالي

شرح الدالة SumProduct

=SUMPRODUCT(A2:A4,B2:B4)

حيث ان المتغير الاول هو المدى A2:A4

و المتغير الثاني B2:B4

و عليه ستكون النتيجة 20

و لكن كيف تم ذلك ؟

سنقوم بتحليل خطوات عمل هذه الدالة

الخطوة الاولى و هي تحويل المدى داخلها الى مصفوفات فيصبح الشكل الداخل لها هكذا

=SUMPRODUCT({3;4;2},{2;1;5})

و من تتم عملية الضرب الداخلية :

=SUMPRODUCT(3*2,4*1,2*5)

و بعد إتمام عملية الضرب يكون الشكل الداخلي

=SUMPRODUCT(6,4,10)

و الخطوة الأخيرة هي عملية الجمع لحاصل عمليات الضرب السابقة فيكون الناتج هو 20.

الى هذا الحد لم نصل الى البداية الحقيقة في التعامل مع هذه الدالة, و لكن قبل أن نبدأ سنعيد عمل الدالة في المثال أعلاه مع تغيير بسيط و هو ان نستخدم علامة الضرب بين المصفوفات بدلا من الفاصلة او الفاصل المنقطة

=SUMPRODUCT(A2:A4*B2:B4)

و في مثالنا أعلاه إستخدمنا الشكل التالي

=SUMPRODUCT(A2:A4,B2:B4)

و الفرق حالياً هو ان المعادلة الاولى تم استخدام متغير واحد بداخله مصفوفتين تم الفصل بينهما بعلامة الضرب اما في المثال الثاني فلقد تم استخدام متغيرين اثنين بمصفوفتين, و لكن هذا الفرق ليس مهماً, حيث سنتطرق الى نقطة اهم من هذه النقطة في الخطوات القادمة, ان شاء الله .

أيضا يمكننا استخدام معادلات الصنف لإستخراج النتائج السابقة و لكننا سنحتاج فقط الدالة sum

حيث سنقوم بكتابة المعادلة التالية

=SUM((A2:A4)*(B2:B4))

و من ثم نقوم بالضغط على Ctrl+Shift+Enter

و سيكون الناتج هو نفسه 20

شرح الدالة SumProduct

و الآن سنرى كيفية استخدام هذه الدالة في عمليات العد

لو افترضنا وجود المدى التالي من A1:A10 و يحتوي الأسماء التالية

الإسم
يحيى
عمر
خبر
يحيى
خبر
عمر
عمر
يحيى
يحيى

و أردنا معرفة عدد مرات تكرار اسم يحيى في هذا المدى، فا إننا نستخدم الدالة COUNTIF

=COUNTIF(A2:A10,"يحيى")

و سيكون الناتج اربع مرات

و لو أردنا استخدام الدالة SumProduct فستكون الدالة بالشكل التالي

=SUMPRODUCT(--(A2:A10="يحيى"))

و لو أردنا استخدام معادلات الصقيف سيكون الشكل التالي للمعادلة

=SUM(IF(A2:A10="1,0,"يحيى"))

و لا ننسى هنا أن نضغط على Ctrl+Shift+Enter

و لكن لنا وقفة هنا مع الدالة SumProduct لماذا إستخدمنا علامتي طرح فيها، لماذا لم يكن شكل المعادلة

=SUMPRODUCT((A2:A10="يحيى"))

قم بتجربة الشكلين

الشكل الاول مع علامتي الطرح – سيعطي النتيجة الصحيحة و هي 4

و لكن الشكل الثاني أعطى نتيجة خاطئة ما هو السبب

و لفهم ذلك سنقوم بتحليل المعادلة، و هذا التحليل مهم جداً أن نفهمه لأن ما سيأتي سيكون مبني عليه :

في المعادلة اعلاه تقوم الدالة SUMPRODUCT

بفحص المدى الذي حددناه و تبحث عن الاسم يحيى فيه، فإن وجدت الاسم ستكون النتيجة True و إن لم تجده ستكون النتيجة False و لو عدنا للمدى أعلاه و طبقنا هذا الكلام سيصبح شكل الجدول كما يلي :

شرح الدالة SumProduct

الإسم	
يحيى	TRUE
عمر	FALSE
خبور	FALSE
يحيى	TRUE
خبور	FALSE
عمر	FALSE
عمر	FALSE
يحيى	TRUE
يحيى	TRUE

و كما هو معروف

ان قيمة True تساوي 1

في حين أن قيمة False تساوي 0

و لكن هنا نتوقف الدالة SumProduct عن قدرتها في تحويل هذه القيمة المنطقية الى قيم رقمية و لن تكون لديها القدرة على جمع هذه النتائج كقيمة منطقية, حيث ستصبح المعادلة هكذا :

=SUMPRODUCT({TRUE;FALSE;FALSE;TRUE;FALSE;FALSE;FALSE;TRUE;TRUE})

و هذا يفسر الحالة الثانية التي عجزت فيها الدالة بدون استخدام علامتي الطرح عن استخراج النتيجة الصحيحة , و لكن ما زال السؤال قائماً ما هي وظيفة علامتي الطرح .

السبب : هو حتى يتم إجبار الاكسيل على تحويل القيم المنطقية الى قيم رقمية, تذكر جيداً كلمة إجبار , و عليه تصبح المصفوفة كما يلي :

الإسم		
يحيى	TRUE	1
عمر	FALSE	0
خبور	FALSE	0
يحيى	TRUE	1
خبور	FALSE	0
عمر	FALSE	0
عمر	FALSE	0
يحيى	TRUE	1
يحيى	TRUE	1

و هنا ستقوم الدالة بجمع قيم رقمية و سيصبح شكلها كما يلي :

=SUMPRODUCT({1;0;0;1;0;0;0;1;1})

و لكن السؤال الذي يطرح نفسه هل هذه هي طرق الإجبار فقط, أم أن هناك طرقاً أخرى يمكن إستخدامها ؟

و الجواب : يوجد هناك عدة طرق أخرى منها :

شرح الدالة SumProduct

1. ضرب المصفوف بالرقم واحد :
=SUMPRODUCT(1*(A2:A10="يحيى"))
2. رفع القيمة للأس 1 :
=SUMPRODUCT((A2:A10="يحيى")^1)
3. جمع القيمة صفر الى المصفوفة:
=SUMPRODUCT(0+(A2:A10="يحيى"))
4. او استخدام الدالة N :
=SUMPRODUCT(N((A2:A10="يحيى")))
5. او ضرب المصفوفة بالدالة True :
=SUMPRODUCT(TRUE*((A2:A10="يحيى")))
6. و طريقة – التي شرحناها سابقاً :
=SUMPRODUCT(--(A2:A10="يحيى"))

إستخدام الدالة SumProduct في العد لأكثر من شرط

لو إفتراضنا وجود البيانات التالية

الاسم	المنطقة
يحيى	الشمال
عمر	الشمال
خبور	الشمال
يحيى	الغرب
خبور	الغرب
عمر	الشمال
عمر	الغرب
يحيى	الشمال
يحيى	الشمال

و أردنا معرفة عدد المرات التي تطابق فيها إسم يحيى و المنطقة الشمالية, أي أن العد هنا بشرطين, و هما :

الشرط الاول : وجود إسم يحيى في المدى الاول A2:A10.

الشرط الثاني : بعد تحقق الشرط الاول ان يقتصر بوجود المنطقة الشمالية في المدى B2:B10.

و الحل : لمن يستخدم الاكسيل 2007 هو استخدام الدالة COUNTIFS, و تكون بالشكل التالي :

=COUNTIFS(A2:A10,"يحيى",B2:B10,"الشمال")

و ستكون النتيجة هي 3

و لكن يمكن الحل أيضاً بإستخدام الدالة SumProduct كما يلي :

=SUMPRODUCT(--(A2:A10="يحيى"),--(B2:B10="الشمال"))

و هي ما يهمنا الآن .

شرح الدالة SumProduct

و لفهم آلية عملها سنقوم بتحليل خطوات الحل :

سنقوم الدالة أولاً بعملية فحص في المدى الاول و البحث عن اسم يحيى و حيثما وجد ستعيد الدالة النتيجة True

و ستعيد المصفوفة التالية من نتائج الفحص المنطقية :

{TRUE;FALSE;FALSE;TRUE;FALSE;FALSE;FALSE;TRUE;TRUE}

و عند وجود علامتي الطرح – سيتم تحويل هذه النتائج الى قيم رقمية كما يلي :

{1;0;0;1;0;0;0;1;1}

و من ثم يتم فحص المدى الثاني و البحث عن كلمة الشمال و حيثما وجدت ستعيد الدالة النتيجة True

و سستكون لدينا المصفوفة التالية :

{TRUE;TRUE;TRUE;FALSE;FALSE;TRUE;FALSE;TRUE;TRUE}

و عند وجود علامتي الطرح – سيتم تحويل هذه النتائج الى قيم رقمية كما يلي :

{1;1;1;0;0;1;0;1;1}

و بذلك يصبح شكل المعادلة النهائي كما يلي :

=SUMPRODUCT({1;0;0;1;0;0;0;1;1},{1;1;1;0;0;1;0;1;1})

و هنا وصلنا للخطوة النهائية في عملية الضرب و التجميع, حيث سيتم ضرب كل رقم بالقيمة التي تقابله , و لفهم ذلك سأقوم بوضع القيم في خلايا مقابله للجدول الرئيسي :

الاسم	المنطقة	الاسم	المنطقة	حاصل عملية الضرب
يحيى	الشمال	1	1	1
عمر	الشمال	0	1	0
خبور	الشمال	0	1	0
يحيى	الغرب	1	0	0
خبور	الغرب	0	0	0
عمر	الشمال	0	1	0
عمر	الغرب	0	0	0
يحيى	الشمال	1	1	1
يحيى	الشمال	1	1	1

لاحظ هنا ان العمود الذي يحتوي الاسم و الذي هو العمود C به الرقم 1 لانه وجد الاسم يحيى و أيضاً العمود الثاني للمنطقة و الذي يمثل العمود D به الرقم 1 لان القيمة تحتوي كلمة الشمال اما العمود الأخير فهو حاصل ضرب 1*1 و تساوي 1

السطر الذي يليه لم يجد الاسم يحيى و كانت النتيجة صفر في حين انه وجد كلمة الشمال فتكون النتيجة 1 و عليه حاصل عملية الضرب 0*1 و تساوي 0 , و هكذا لباقي قيم الجدول , و بناء على الجدول أعلاه ستكون الصورة النهائية للمعادلة بعد عملية الضرب المتقابل و التي ستعطي النتيجة 3 , كما يلي :

=SUMPRODUCT({1;0;0;0;0;0;0;1;1})

شرح الدالة SumProduct

و أيضا نفس الخطوات في حل وجد شروط أخرى, فعلى سبيل المثال في وجود هذه البيانات :

الإسم	المنطقة	المدينة
يحيى	الشمال	عمان
عمر	الشمال	عمان
خبور	الشمال	العقبة
يحيى	الغرب	العقبة
خبور	الغرب	عمان
عمر	الشمال	عمان
عمر	الغرب	عمان
يحيى	الشمال	عمان
يحيى	الشمال	العقبة

و أردنا معرفة عدد مرات تواجد يحيى في المنطقة الشمالية في مدينة عمان, بمعنى أنه وجد ثلاث شروط, و هي :

الشرط الاول : اسم يحيى في المدى الاول .

الشرط الثاني : بعد تحقق الشرط الاول أن تكون في المنطقة الشمالية .

الشرط الثالث : بعد تحقق الشرطين أعلاه أن تكون للمدينة عمان .

كما ذكرنا سابقاً في الاكسيل 2007 يمكننا استخدام الدالة COUNTIFS, بالشكل التالي :

=COUNTIFS(A2:A10,"يحيى",B2:B10,"الشمال",C2:C10,"عمان")

و أيضا يمكننا ذلك مع الدالة SumProduct, كما يلي :

=SUMPRODUCT(--(A2:A10="يحيى"),--(B2:B10="الشمال"),--(C2:C10="عمان"))

و بإتباع خطوات التحليل أعلاه سنصل إلى فكرة عملها مع الثلاثة شروط.

و أيضاً نفس خطوات الحل مع الأربعة شروط

الإسم	المنطقة	المدينة	اسم الصنف
يحيى	الشمال	عمان	موبايل
عمر	الشمال	عمان	موبايل
خبور	الشمال	العقبة	كمبيوتر
يحيى	الغرب	العقبة	كمبيوتر
خبور	الغرب	عمان	كمبيوتر
عمر	الشمال	عمان	موبايل
عمر	الغرب	عمان	كمبيوتر
يحيى	الشمال	عمان	كمبيوتر
يحيى	الشمال	العقبة	موبايل

شرح الدالة SumProduct

لو أردنا معرفة عدد مرات مبيعات يحيى في المنطقة الشمالية في المديق عمان للصنف الموبايل , يمكننا استخدام الدالة COUNTIFS بالشكل التالي :

=COUNTIFS(A2:A10,"يحيى",B2:B10,"الشمال",C2:C10,"عمان",D2:D10,"موبايل")

و ستكون النتيجة رقم 1

و ايضاً مع الدالة SumProduct سيكون شكل المعادلة كما يلي :

=SUMPRODUCT(--(A2:A10="يحيى"),--(B2:B10="الشمال"),--(C2:C10="عمان"),--(D2:D10="موبايل"))

و ستكون النتيجة 1

و هكذا نكون قد تعرفنا على قدرة هذه الدالة في التعامل مع عمليات العد المشروطة, و لكن بقي نقطة واحدة أن جميع المعادلة أعلاه استخدمت الشرط المنطقي AND بمعنى لاحظ في حالة الجمع لشرطين :

يحيى And الشمال

و في حالة الثلاث شروط

يحيى And الشمال And عمان

و في حالة الأربع شروط

يحيى And الشمال And عمان And موبايل

و من ناحية منطقية بما اننا استخدمنا And إذا يمكننا استخدام OR

و لكن كيف سيتم ذلك :

بالعودة إلى المثال الأول :

و استخدام نفس المعلومات

الاسم	المنطقة
يحيى	الشمال
عمر	الشمال
خبور	الشمال
يحيى	الغرب
خبور	الغرب
عمر	الشمال
عمر	الغرب
يحيى	الشمال
يحيى	الشمال

طلبنا يحيى And الشمال

و رأينا كيف أنه أمكننا عمل ذلك, و لكن نريد هنا أن نرى عدد مرات تكرار يحيى أو الشمال

شرح الدالة SumProduct

يحيى Or الشمال

و الفرق سيكون بسيط فقط سنبتدل الفاصل بعلامة +

ليصبح شكل المعادلة :

=SUMPRODUCT((A2:A10="يحيى")+(B2:B10="الشمال"))

و ستكون النتيجة 10

و لكن لماذا لم نستخدم علامتي الطرح - -

و للإجابة على ذلك سنعود للوراء إلى عمليات إجبار تحويل القيم المنطقية إلى قيم رقمية و قمت بذكر ستة طرق و بقي طريقة واحد تستخدم في حالة توفر أكثر من شرط و هي إستبدال الفاصلة بعلامة الضرب فلقد استخدمنا سابقاً هذه المعادلة مع وجود علامتي الطرح - -

=SUMPRODUCT(--(A2:A10="يحيى"),--(B2:B10="الشمال"))

يمكن إستبدالها بالمعادلة التالية :

=SUMPRODUCT((A2:A10="يحيى")*(B2:B10="الشمال"))

و بناء على ذلك يمكننا العودة إلى جميع الأمثلة السابقة و استبدال علامتي الطرح و الفاصلة بعلامة الضرب فقط

ملاحظة : يوجد فرق بين علامة الضرب و الفاصلة في طريقة أداء عمل المعادلة و هذا سنصل إليه في الشروحات القادمة و ليس الآن .

و لتحليل المعادلة أعلاه في عملية جمع نتائج العمليات الشرطية سنرى ذلك بخطوات سريعة:

فلقد قامت الدالة بعملية بحث عن الشروط المنطقية في المدى الأول و الثاني و بناء مصفوفتين مستقلتين و قد شرحنا سابقاً كيفية تحويلهما في الأكسيل و من ثم قامت بعملية الجمع للشروط المتحققة و التي تقابلها القيمة True, و أصبح الدالة تفسر العبارة التالي: عدد مبيعات يحيى أو OR المنقطة الشمالية , يحيى وجد أربع مرات في المدى الأول و منطقة الشمال وجدت ستة مرات و حاصل جمعهما يساوي عشرة .

و نفس الخطوات في حال وجد أكثر من شرطين

فعند وجود ثلاثة شروط :

يحيى Or الشمال Or عمان

نستخدم الدالة التالية :

=SUMPRODUCT(--(A2:A10="يحيى")+(B2:B10="الشمال")+(C2:C10="عمان"))

شرح الدالة SumProduct

و عند وجود أربعة شروط

يحيى Or الشمال Or عمان Or موبائل

=SUMPRODUCT((A2:A10="يحيى")+(B2:B10="الشمال")+(C2:C10="عمان")+(D2:D10="موبائل"))

و لكن في الحياة العملية لا نحتاج هكذا شروط بل قد نحتاج شروط مدمجة

كان نقول

مبيعات يحيى في المنطقة الشمالية أو مدينة عمان

يحيى And الشمال Or عمان

يمكننا عمل ذلك بالمعادلة التالية :

=SUMPRODUCT((A2:A10="يحيى")*(B2:B10="الشمال")+(C2:C10="عمان"))

و سيكون الناتج 9

و لو أردنا شرط أكثر تعقيد

يحيى And الشمال Or عمان And موبائل

=SUMPRODUCT((((A2:A10="يحيى")*(B2:B10="الشمال"))+(C2:C10="عمان")*(D2:D10="موبائل")))

و سيكون الناتج 6

و بناء عليه نستنتج

أن عملية الجمع تشير إلى الشرط المنطقي OR

و أن عملية الضرب تشير إلى الشرط المنطقي AND